
Guía 8 - Ejercicios adicionales para el primer parcial

Programación con Objetos 1

10/08/2018

1. Method Lookup

Ejercicio 1: Virus

A partir de las siguientes definiciones de clase, y considerando las variables definidas en el test

```
1  class Virus {
2    var potencia = 1000
3    var cantidadDeInfectados = 0
4
5    method infectar(cuerpo) {
6      potencia -= cuerpo.peso()
7      cantidadDeInfectados += 1
8    }
9
10   method potencia() {
11     return potencia
12   }
13
14   method tieneExperiencia() {
15     return cantidadDeInfectados > 3
16   }
17
18   method quiereInfectar() {
19     return potencia > 500
20   }
21 }
22
23
24 class VirusGordito inherits Virus {
25   var peso = 20
26   override method infectar(cuerpo) {
27     peso += cuerpo.peso() / 100.0
28   }
29
30   method peso() {
31     return peso
32   }
33 }
34
35
36 class VirusRechonchito inherits VirusGordito {
37   override method quiereInfectar() {
38     return super() and self.peso() > 100
39   }
40 }
41
42 class VirusConLiquido inherits Virus {
43   var dosisLiquido = 20
44   override method infectar(cuerpo) {
45     super(cuerpo)
```

```
46     dosisLiquido -= 1
47   }
48
49   method comerMiel() {
50     dosisLiquido += 15
51   }
52 }
53
54 class VirusConLiquidoDenso inherits VirusConLiquido {
55   var densidad = 10
56   override method infectar(cuerpo) {
57     super(cuerpo)
58     densidad -= 2
59   }
60
61   override method comerMiel() {
62     super()
63     densidad += 5
64   }
65
66   method estaRobusto() {
67     return densidad > 10 and self.potencia() > 2000
68   }
69
70   override method quiereInfectar() {
71     return super() and densidad > 4
72   }
73 }
74
75
76 class VirusMusculoso inherits VirusConLiquidoDenso {
77   override method potencia() {
78     return super() + 500
79   }
80 }
81
82 class Ornitorrinco {
83   method peso() {
84     return 40
85   }
86 }
```

Listing 1: methodLookupVirus.wlk

```
1 test "ejemplos de virus" {
2   var virus1 = new VirusGordito()
3   var virus2 = new VirusRechonchito()
4   var virus3 = new VirusConLiquido()
5   var virus4 = new VirusConLiquidoDenso()
6   var pepe = new Ornitorrinco()
7 }
```

Listing 2: methodLookupVirusTest.wtest

Se pide:

1. Indicar qué atributos tiene cada uno de los cuatro virus creados en el test.
2. Indicar cuáles de ellos cambian su valor si a cada uno le envió el mensaje `infectar(pepe)`.
3. Supongamos que `virus4` tiene 8 de densidad, 10 dosis de líquido y 300 de potencia. ¿Quiere infectar? Explicar por qué.

4. Agregamos este método en `VirusConLiquido`
`method quiereInfectar() = dosisLiquido > 5`
¿cómo cambia la respuesta del punto anterior?
5. Para cada uno de estos mensajes, indicar cuáles de los cinco objetos los entienden
 - a) `comerMiel`
 - b) `quiereInfectar`
 - c) `peso`
 - d) `tieneExperiencia`
 - e) `estaRobusto`
6. Con el código original, supongamos que la potencia de `virus3` es 150 y de los otros tres es 3000, el peso de `virus1` y `virus2` es 40, y la densidad de `virus4` es 35. Si le pregunto a cada uno si `quiereInfectar`, ¿qué me responde?
7. Con el código original, agreguemos esta variable al test
`var virus5 = new VirusMusculoso()`
y supongamos que tanto `virus4` como `virus5` tienen: 50 como valor del atributo `densidad`, y 1800 como valor del atributo `potencia`. Si les pregunto a `virus4` y a `virus5` si están robustos, ¿qué me responde en cada caso? Explicar.
8. Agregar las siguientes variantes de virus
 - a) `VirusConLiquidoSabio`, es como los virus líquidos, con la única diferencia que si tiene más de 50 dosis de líquido entonces tiene experiencia, independientemente de la cantidad de infectados. Si no llega a 50 dosis, entonces sí corre la condición que trae de la clase `Virus`.
 - b) `VirusFiaca`: nunca quiere infectar.

Ejercicio 2: Civilizaciones

El código que sigue define las clases `Civilizacion`, `CivilizacionComercial`, `CivilizacionComercialSuper`, `CivilizacionCultural`, `CivilizacionMilitar` y `CivilizacionPoetica`, además del objeto `juego`.

```
1 class Civilizacion {
2     var _puntajeCientifico
3     var _puntajeMilitar
4     var _puntajeEconomico
5
6     constructor(cien, mil, eco) {
7         _puntajeCientifico = cien
8         _puntajeMilitar = mil
9         _puntajeEconomico = eco
10    }
11
12    method puntaje(){
13        return self.puntajeCientifico()
14        + self.puntajeMilitar() + self.puntajeEconomico()
```

```
15     }
16
17     method puntajeCientifico() {
18         return _puntajeCientifico
19     }
20     method puntajeMilitar() {
21         return _puntajeMilitar
22     }
23     method puntajeEconomico() {
24         return _puntajeEconomico
25     }
26 }
27
28 class CivilizacionComercial inherits Civilizacion {
29     var _cantTratadosComerciales
30     var _cantEscuelasContables
31
32     constructor(cien,mil,eco,tratadosCom,escuelasContables) =
33     super(cien,mil,eco) {
34         _cantTratadosComerciales = tratadosCom
35         _cantEscuelasContables = escuelasContables
36     }
37
38     override method puntajeEconomico() {
39         return super()
40             + self.cantTratadosComerciales() * self.baseComercio()
41     }
42
43     override method puntajeCientifico() {
44         return super()
45             + self.cantEscuelasContables() * self.baseComercio()
46     }
47     method baseComercio() {
48         return juego.baseComercio()
49     }
50
51     method cantTratadosComerciales() {
52         return _cantTratadosComerciales
53     }
54     method cantEscuelasContables() {
55         return _cantEscuelasContables
56     }
57 }
58
59
60 class CivilizacionComercialSuper inherits CivilizacionComercial {
61     constructor(cien,mil,eco,tratadosCom,escuelasContables) =
62     super(cien,mil,eco,tratadosCom,escuelasContables)
63
64     override method baseComercio() {
65         return super() * 2
66     }
67 }
68
69 class CivilizacionCultural inherits Civilizacion {
70     var _cantMuseos
71     var _cantUniversidades
72
73     constructor(cien,mil,eco,museos,univs) =
74     super(cien,mil,eco) {
75         _cantMuseos = museos
76         _cantUniversidades = univs
77     }
78
79     override method puntaje() {
80         return super() + self.puntajeCultural()
81     }
82
83     method puntajeCultural() {
84         return self.cantMuseos() * juego.baseCultural()
85     }
86 }
```

```
86
87
88     override method puntajeCientifico() {
89         return super()
90             + self.cantUniversidades() * juego.baseCultural()
91     }
92
93     method cantMuseos() {
94         return _cantMuseos
95     }
96
97     method cantUniversidades() {
98         return _cantUniversidades
99     }
100 }
101
102 class CivilizacionMilitar inherits Civilizacion {
103     constructor(cien,mil,eco) = super(cien,mil,eco)
104
105     override method puntajeCientifico() {
106         return 0
107     }
108     override method puntajeEconomico() {
109         return 0
110     }
111 }
112
113 class CivilizacionPoetica inherits CivilizacionCultural {
114     var _nombreLiterario
115
116     constructor(cien,mil,eco,museos,univs,nombre) =
117     super(cien,mil,eco,museos,univs) {
118         _nombreLiterario = nombre
119     }
120
121     override method puntaje() {
122         return self.puntajeCultural()
123             + self.nombreLiterario().size() * 5
124     }
125
126     method nombreLiterario() {
127         return _nombreLiterario
128     }
129 }
130
131 object juego {
132     var _baseComercio = 0
133     var _baseCultural = 0
134
135
136     method baseComercio(unValor) {
137         _baseComercio = unValor
138     }
139
140     method baseComercio() {
141         return _baseComercio
142     }
143
144     method baseCultural(unValor) {
145         _baseCultural = unValor
146     }
147
148     method baseCultural() {
149         return _baseCultural
150     }
151 }
```

Listing 3: methodLookupCiv.wlk

Se pide:

a)

Armar el diagrama de clases que incluya las 6 clases definidas, indicando la herencia y qué métodos se definen en cada una.

b)

Indicar qué valores hay que poner en reemplazo de fenT, fenC, carT, carC, milT, milC, espT, espC, delT, delC para que el siguiente test dé verde.

```

1  test "puntajes" {
2      juego.baseComercio(5)
3      juego.baseCultural(8)
4
5      var fenicia = new CivilizacionComercial(
6          20, 15, 100,    // científico, militar, economico
7          5, 6           // tratadosComerciales, escuelasContables
8      )
9
10     var cartago = new CivilizacionComercialSuper(
11         20, 35, 100,    // científico, militar, economico
12         10, 4           // tratadosComerciales, escuelasContables
13     )
14
15     var mileto = new CivilizacionCultural (
16         40, 50, 20,     // científico, militar, economico
17         6, 9            // museos, universidades
18     )
19
20
21     var esparta = new CivilizacionMilitar (
22         5, 200, 45      // científico, militar, economico
23     )
24
25     var delfos = new CivilizacionPoetica (
26         60, 2, 20,     // científico, militar, economico
27         10, 2,         // museos, universidades
28         "Delfos la bella" // nombre, tiene 15 letras
29     )
30
31     assert.equals(/* fenT */, fenicia.puntaje())
32     assert.equals(/* fenC */, fenicia.puntajeCientifico())
33     assert.equals(/* carT */, cartago.puntaje())
34     assert.equals(/* carC */, cartago.puntajeCientifico())
35     assert.equals(/* milT */, mileto.puntaje())
36     assert.equals(/* milC */, mileto.puntajeCientifico())
37     assert.equals(/* espT */, esparta.puntaje())
38     assert.equals(/* espC */, esparta.puntajeCientifico())
39     assert.equals(/* delT */, delfos.puntaje())
40     assert.equals(/* delC */, delfos.puntajeCientifico())
41 }

```

Listing 4: methodLookupCivTest.wtest

c)

Indicar cómo implementar la clase `CivilizacionMilitar` si se quiere preservar los puntajes científico y económico, pero que no influyan en el puntaje total. P.ej. si queremos que en el test anterior, el puntaje total de Esparta sea el mismo que se calculó, pero el puntaje científico sea 5 y el económico 45.

Respuestas del punto b)

Fenicia	total	190
Fenicia	científico	50
Cartago	total	295
Cartago	científico	60
Mileto	total	230
Mileto	científico	112
Esparta	total	200
Esparta	científico	0
Delfos	total	155
Delfos	científico	76

Ejercicio 3: El escritor automático

El código que sigue define las clases YoAprendi, YoAprendiConfigurable, YoFui, YoFuiDeportivo, YoAprendiEnEscuela, AprendiBaile, AprendiBienBaile, AprendiSalsa, AprendiSalsaLunes y Escuela.

```

1  class YoAprendi {
2      method accion()
3      method objeto()
4      method final()
5      method encabezado() {
6          return "yo aprendi a"
7      }
8
9      method frase() {
10         return self.encabezado() + " "
11         + self.accion() + " "
12         + self.objeto() + " "
13         + self.final()
14     }
15 }
16
17 class YoAprendiConfigurable inherits YoAprendi {
18     var _accion
19     var _objeto
20     var _final
21
22     constructor(accion,objeto,final) {
23         _accion = accion
24         _objeto = objeto
25         _final = final
26     }
27
28     override method accion() { return _accion }
29     override method objeto() { return _objeto }
30     override method final() { return _final }
31 }
32
33 class YoFui inherits YoAprendiConfigurable {
34     constructor(accion,objeto,final) =
35         super(accion,objeto,final)
36
37     override method encabezado() { return "yo fui a" }
38 }
39
40 class YoFuiDeportivo inherits YoFui {
41     constructor(accion,objeto,final) =
42         super(accion,objeto,final)
43

```

```

44     override method accion() { return super() + " con energia" }
45     override method final() { return super() + ", y me gusto" }
46 }
47
48 class YoAprendiEnEscuela inherits YoAprendi {
49     var _escuela
50     constructor(escuela) { _escuela = escuela }
51
52     override method accion() {
53         return self.escuela().queEnseniaAHacer()
54     }
55     override method objeto() { return self.escuela().objeto() }
56     override method encabezado() { return "les cuento que" }
57     override method final() {
58         return "en la escuela " + self.escuela().nombre()
59     }
60     method escuela() = { return _escuela }
61 }
62 class AprendiBaile inherits YoAprendi {
63     override method accion() { return "bailar" }
64 }
65
66 class AprendiBienBaile inherits AprendiBaile {
67     override method accion() {
68         return super() + " con precision"
69     }
70 }
71
72 class AprendiSalsa inherits AprendiBienBaile {
73     var _cuando
74     constructor(cuando) { _cuando = cuando }
75
76     override method objeto() { return "salsa" }
77     override method final() {
78         return self.cuando() + ", y me fue bien"
79     }
80     method cuando() { return _cuando }
81 }
82
83 class AprendiSalsaLunes inherits AprendiSalsa {
84     constructor(cuando) = super(cuando)
85
86     override method cuando() { return super() + " los lunes" }
87 }
88
89 class Escuela {
90     var _nombre
91     var _queEnseniaAHacer
92     var _objeto
93
94     constructor(nombre, queEnsenia, objeto) {
95         _nombre = nombre
96         _queEnseniaAHacer = queEnsenia
97         _objeto = objeto
98     }
99
100     method nombre() { return _nombre }
101     method queEnseniaAHacer() { return _queEnseniaAHacer }
102     method objeto() { return _objeto }
103 }

```

Listing 5: methodLookupFrases.wlk

Se pide:

a)

Armar el diagrama de clases que incluya las 10 clases definidas, indicando la herencia y qué métodos se definen en cada una, y además, cuáles de estas clases son abstractas.

b)

Indicar qué valores hay que poner en reemplazo de `sui`, `nadar`, `nadarC`, `ing`, `salsi`, `salsiL` para que el siguiente test dé verde.

```

1  test "frases" {
2      var icana = new Escuela(
3          "ICANA", "hablar", "ingles"      // nombre, accion, objeto
4      )
5
6      var sui = new YoAprendiConfigurable(
7          "ser", "formal y cortés",      // accion, objeto
8          "cortandome el pelo una vez por mes" // final
9      )
10
11     var nadar = new YoFui(
12         "nadar", "crawl", "al rio"      // accion, objeto, final
13     )
14
15     var nadarCopado = new YoFuiDeportivo(
16         "nadar", "crawl", "al rio"      // accion, objeto, final
17     )
18
19     var inglich = new YoAprendiEnEscuela(icana) // escuela
20
21     var salsita = new AprendiSalsa(
22         "el mes pasado" // cuando
23     )
24
25     var salsitaLunes = new AprendiSalsaLunes (
26         "el mes pasado" // cuando
27     )
28
29     assert.equals(/* sui */, sui.frase())
30     assert.equals(/* nadar */, nadar.frase())
31     assert.equals(/* nadarC */, nadarCopado.frase())
32     assert.equals(/* ing */, inglich.frase())
33     assert.equals(/* salsi */, salsita.frase())
34     assert.equals(/* salsiL */, salsitaLunes.frase())
35 }

```

Listing 6: methodLookupFrasesTest.wtest

c)

Definir la clase `AprendiTango`, que a partir de una acción complementaria y un lugar, arme la frase

“Yo aprendi a bailar y <<acción complementaria>> tango en <<lugar>>.”

P.ej. si definimos como acción complementaria “disfrutar” y como lugar “El Beso”, la frase queda:

“Yo aprendi a bailar y disfrutar tango en El Beso.”

Armar `AprendiTango` a partir de las clases que están implementadas, separando en acción (que incluye siempre “bailar y” más la complementaria), objeto (que es tango), y final (que es el lugar con “en” adelante).

Ejercicio 4: Trace

A partir de las siguientes clases:

```
1  class A {
2
3      method m1() { return "A.m1" }
4
5      method m2() { return "A.m2" }
6
7      method m3() { return "A.m3" }
8  }
9
10 class B inherits A {
11
12     override method m1() { return "B.m1 " + super() }
13
14     override method m2() { return "B.m2 " + self.m3() }
15
16     method m4() { return "B.m4" }
17 }
18
19 class C inherits B {
20
21     override method m1() { return "C.m1" }
22
23     override method m2() {
24         return "C.m2 " + super() + " " + self.m4()
25     }
26
27     override method m3() { return "C.m3" }
28 }
```

Listing 7: methodLookupTrace.wlk

Indicar que devuelve cada línea de código:

```
1  var o = new A()
2      o.m1()
3      o.m2()
4      o.m3()
5      o.m4()
6
7      o = new B()
8      o.m1()
9      o.m2()
10     o.m3()
11     o.m4()
12
13     o = new C()
14     o.m1()
15     o.m2()
16     o.m3()
17     o.m4()
```

Listing 8: methodLookupTrace.wtest

Ejercicio 5: Trace 2

Dadas las siguientes clases

```
1 class Abuela {
2
3     method m1() {
4         return "A.m1 " + self.m2() + " " + self.m3()
5     }
6
7     method m2() {
8         return "A.m2"
9     }
10    //abstracto
11    method m3()
12 }
13
14 class Madre inherits Abuela {
15     override method m2() {
16         return "M.m2 " + super() + " " + self.m3()
17     }
18
19     override method m3() {
20         return "M.m3"
21     }
22 }
23
24 class Hija inherits Madre {
25
26     override method m1() {
27         return "H.m1 " + self.m3() + " " + super()
28     }
29
30 }
```

Listing 9: methodLookupTrace2.wlk

Indique que devuelve cada línea de código:

```
1     var p
2
3     p = new Madre()
4     p.m1()
5     p.m2()
6     p.m3()
7
8     p = new Hija()
9     p.m1()
10    p.m2()
11    p.m3()
```

Listing 10: methodLookupTrace2.wpgm

Ejercicio 6: Gritos

A partir de las siguientes clases:

```

1  class Gritador {
2      method gritar() {
3          return self.gritito() + self.granGrito() + self.final()
4      }
5
6      method gritito() { return "uy" }
7
8      method granGrito() { return "UY" }
9
10     method final() { return "!!" }
11 }
12
13 class Gritador1 inherits Gritador{
14     override method gritito() { return "ay" }
15
16     override method granGrito() {
17         return self.gritito() + "AAH"
18     }
19 }
20
21 class Gritador2 inherits Gritador1 {
22
23     override method gritito() { return "aa" + super() }
24
25     override method final() {
26         return self.granGrito() + super()
27     }
28
29     override method gritar() {
30         return self.final() + self.granGrito() + self.gritito()
31     }
32 }
33
34 class Gritador3 inherits Gritador2 {
35
36     override method granGrito() { return "EEE" }
37
38     override method final() { return self.gritito() }
39 }

```

Listing 11: methodLookupGritos.wlk

Indicar que devuelve cada línea de código:

```

1  new Gritador().gritar()
2  new Gritador1().gritar()
3  new Gritador2().gritar()
4  new Gritador3().gritar()

```

Listing 12: methodLookupGritos.wpgm

De este te damos las **respuestas**

```

Gritador    uyUY!!
Gritador1   ayayAAH!!
Gritador2   aaayAAH!!aaayAAHaaay
Gritador3   aaayEEEaaay

```

Ejercicio 7: Vikingos

Dadas las siguientes clases

```

1 class Vikingo {
2
3   method accionesDeSaqueo() {
4     return self.accionesTesoros() + self.accionesEnemigos()
5   }
6   method accionesTesoros() {
7     return ["saquear tesoros", "fundir oro"]
8   }
9   method accionesEnemigos() { return ["matar guerreros"] }
10 }
11 class Conde inherits Vikingo {
12
13   override method accionesEnemigos() {
14     return super() + ["matar familia"]
15   }
16   override method accionesDeSaqueo() {
17     return super() + self.accionesLugar()
18   }
19   method accionesLugar() { return ["incendiar lugar"] }
20 }
21 class Rey inherits Conde{
22
23   override method accionesEnemigos() {
24     return ["reclutar jefes vencidos"] + super()
25   }
26   override method accionesLugar() {
27     return super() + ["echar sal", "arar"]
28   }
29   override method accionesTesoros() { return ["dividir tesoros"]}
30 }
31 class Profeta inherits Vikingo {
32
33   override method accionesTesoros() { return [] }
34
35   override method accionesEnemigos() {
36     return ["predecir futuro de enemigos"]
37   }
38 }

```

Listing 13: methodLookupVikingos.wlk

Indique que devuelve cada línea de código:

```

1 new Vikingo().accionesDeSaqueo()
2 new Conde().accionesDeSaqueo()
3 new Rey().accionesDeSaqueo()
4 new Profeta().accionesDeSaqueo()

```

Listing 14: methodLookupVikingos.wpgm

Nota: Recordar que las listas entienden el mensaje + para concatenar. El resultado es una nueva lista. P.ej. el resultado de

```
["alfa","beta"] + ["gamma","iota","epsilon"] + [] + ["omega"] es
["alfa","beta","gamma","iota","epsilon","omega"]
```

Ejercicio 8: Skills

Dadas las siguientes clases

```

1  class Junior {
2      method actividades() {
3          return self.burocracia() + self.programacion()
4      }
5      method burocracia() {
6          return ["cargar horas"]
7      }
8      method programacion() {
9          return ["resolver issues"]
10     }
11 }
12 class SemiSenior inherits Junior {
13     override method programacion() {
14         return super() + ["detectar issues"]
15     }
16     override method actividades() {
17         return super() + self.planificacion()
18     }
19     method planificacion() {
20         return ["estimar issues"]
21     }
22 }
23 class Senior inherits SemiSenior {
24     override method programacion() {
25         return ["refactorizar arquitectura"] + super()
26     }
27     override method planificacion() {
28         return ["coordinar reunion"] + self.demos()
29     }
30     method demos() {
31         return ["coordinar demo"]
32     }
33     override method burocracia() {
34         return []
35     }
36 }
37 class Ninja inherits Senior {
38     override method programacion() {
39         return ["resolver todo"]
40     }
41     override method demos() {
42         return ["salvar demo"]
43     }
44 }
45 class Acomodado inherits Junior { //OJO!! Mira bien de quien hereda!!
46     override method programacion() {
47         return []
48     }
49     override method burocracia() {
50         return ["tomar cafe"] + super()
51     }
52 }

```

Listing 15: methodLookupSkills.wlk

Indique que devuelve cada línea de código:

```

1  new Junior().actividades()
2  new SemiSenior().actividades()
3  new Senior().actividades()
4  new Ninja().actividades()
5  new Acomodado().actividades()

```

Listing 16: methodLookupSkills.wpgm

Ejercicio 9: Jedi

Dadas las siguientes clases

```

1  class Personaje {
2
3      method habilidades() {
4          return self.batalla() + self.extras()
5      }
6      method batalla() {
7          return ["disparar laser"]
8      }
9      method extras() {
10         return ["programar droide"]
11     }
12 }
13 class Padawan inherits Personaje {
14
15     override method batalla() {
16         return ["usar sable laser"]
17     }
18     override method extras() {
19         return ["sentir la fuerza"] + super()
20     }
21 }
22 class Jedi inherits Padawan {
23
24     override method habilidades() {
25         return super() + self.estrategia()
26     }
27     method estrategia() {
28         return ["comandar clones"]
29     }
30     override method extras() {
31         return ["construir sable", "mover objetos"] + super()
32     }
33 }
34 class MaestroJedi inherits Jedi {
35
36     override method estrategia() {
37         return self.jedis() + super()
38     }
39     method jedis() {
40         return ["comandar jedis"]
41     }
42 }
43 class LordSith inherits MaestroJedi {
44
45     override method habilidades() {
46         return self.estrategia() + ["usar lado oscuro"]
47     }
48     override method jedis() {
49         return ["matar jedis"]
50     }
51 }

```

Listing 17: methodLookupJedi.wlk

Indique que devuelve cada línea de código:

```

1  new Personaje().habilidades()
2  new Padawan().habilidades()
3  new Jedi().habilidades()
4  new MaestroJedi().habilidades()
5  new LordSith().habilidades()

```

Listing 18: methodLookupJedi.wpgm

Ejercicio 10: Soldados

Dadas las siguientes clases

```

1  class Soldado {
2
3      method habilidades() {
4          return self.batalla() + self.extras()
5      }
6      method batalla() {
7          return ["fusil"]
8      }
9      method extras() {
10         return ["limpiar"]
11     }
12 }
13 class Cabo inherits Soldado {
14
15     override method batalla() {
16         return ["artilleria"] + super()
17     }
18     override method extras() {
19         return ["fumar"]
20     }
21 }
22 class Sargento inherits Cabo {
23
24     override method habilidades() {
25         return super() + self.estrategia()
26     }
27     method estrategia() {
28         return ["comandar soldados"]
29     }
30     override method extras() {
31         return ["poker"] + super()
32     }
33 }
34 class Teniente inherits Sargento {
35
36     override method estrategia() {
37         return self.politica() + super()
38     }
39     method politica() {
40         return ["contentar al superior"]
41     }
42 }
43 class Coronel inherits Teniente {
44
45     override method habilidades() {
46         return self.estrategia() + ["planificar"]
47     }
48     override method politica() {
49         return ["negociar"]
50     }
51 }

```

Listing 19: methodLookupSoldados.wlk

Indique que devuelve cada línea de código:

```

1  new Soldado().habilidades()
2  new Cabo().habilidades()
3  new Sargento().habilidades()
4  new Teniente().habilidades()
5  new Coronel().habilidades()

```

Listing 20: methodLookupSoldados.wpgm

2. Polimorfismo

Ejercicio 11: Exámenes

Reorganice y modifique el siguiente código utilizando polimorfismo para mejorar la distribución de responsabilidades. Se permite (de hecho, se alienta a) agregar más clases a las planteadas, pasando parte del código a las nuevas clases.

```

1  class Alumno {
2  method puntaje(examen) {
3      var puntos = 0
4      examen.preguntas().forEach({ preg =>
5          if (preg.tipo() == "Exacta") {
6              if (preg.lePegoJusto()) {
7                  puntos += preg.puntajeQueOtorga()
8              }
9          }
10         if (preg.tipo() == "Aproximada") {
11             if (preg.lePegoJusto()) {
12                 puntos += preg.puntajeQueOtorga()
13             }
14             if (preg.respuestaCorrecta() == (preg.respuesta() - 1)) {
15                 puntos += preg.puntajeQueOtorga() - 2
16             }
17             if (preg.respuestaCorrecta() == (preg.respuesta() + 1)) {
18                 puntos += preg.puntajeQueOtorga() - 2
19             }
20         }
21         // en las preguntas de este tipo, la respuesta
22         // es una lista ordenada
23         if (preg.tipo() == "Listas") {
24             if (preg.lePegoJusto()) {
25                 puntos += preg.puntajeQueOtorga()
26             } else {
27                 if (preg.respuestaCorrecta().asSet() ==
28                     preg.respuesta().asSet()) {
29                     puntos += preg.puntajeQueOtorga() - 5
30                 }
31             }
32         }
33     })
34     if (examen.preguntas().all({preg => preg.contesto()})) {
35         puntos += 10
36     }
37     return puntos
38 }
39 }
40
41 class Examen {
42     var preguntas
43
44     constructor(_preguntas) {preguntas = _preguntas}
45
46     method preguntas() = preguntas
47 }
48
49 class Pregunta {
50     var respuestaCorrecta
51     var respuesta
52     var puntajeQueOtorga
53     var tipo
54
55     constructor(_respuestaCorrecta, _respuesta, _puntajeQueOtorga, _tipo) {
56         respuestaCorrecta = _respuestaCorrecta
57         respuesta = _respuesta
58         puntajeQueOtorga = _puntajeQueOtorga
59         tipo = _tipo

```

```

60 }
61
62 method respuestaCorrecta() {
63     return respuestaCorrecta
64 }
65 method respuesta() {
66     return respuesta
67 }
68 method puntajeQueOtorga() {
69     return puntajeQueOtorga
70 }
71
72 method lePegoJusto() {
73     return self.respuestaCorrecta() == self.respuesta()
74 }
75
76 method tipo() {
77     return tipo
78 }
79 method contesto() {
80     return respuesta != null
81 }
82
83 }

```

Listing 21: polimorfismoExamen.wlk

Ejercicio 12: Renderización de una pantalla

Reorganice y modifique el siguiente código utilizando polimorfismo para mejorar la distribución de responsabilidades. Se permite (de hecho, se alienta a) agregar más clases a las planteadas, pasando parte del código a las nuevas clases.

Tener en cuenta las siguientes aclaraciones:

- Los métodos no desarrollados en la clase `Pantalla` indican que una pantalla sabe trazar una línea en un color a partir de la posición de un cursor, y también mover el cursor. La implementación de estas operaciones no está entre el código que debe reorganizar.
- Obsérvese que para dibujar una figura, antes de empezar hay que prender el cursor, y lo último que hay que hacer es apagarlo. Estas también son operaciones responsabilidad de la pantalla, y que no están incluidas en la reorganización de código.
- Nótese también que cada figura debe ser dibujada en rojo y en verde, esto es cierto tanto para las figuras que han sido implementadas, como para las que pudieran agregarse. También, hay que tener en cuenta que después de dibujar una figura, el cursor debe estar en el mismo lugar en el que estaba antes de empezar a dibujarla.

```

1 class Pantalla {
2     method dibujar(figura) {
3         self.prenderCursor()
4         if (figura.tipo() == "lineaHorizontal") {
5             // en rojo
6             self.lineaDerecha(figura.tamano(), colorRojo)
7             self.moverCursorIzquierda(figura.tamano())
8             // en verde

```

```
9         self.lineaDerecha(figura.tamano(), colorVerde)
10        self.moverCursorIzquierda(figura.tamano())
11    }
12    if (figura.tipo() == "ele") {
13        // en rojo
14        self.lineaAbajo(figura.tamano(), colorRojo)
15        self.lineaDerecha(figura.tamano(), colorRojo)
16        self.moverCursorIzquierda(figura.tamano())
17        self.moverCursorArriba(figura.tamano())
18        // en verde
19        self.lineaAbajo(figura.tamano(), colorVerde)
20        self.lineaDerecha(figura.tamano(), colorVerde)
21        self.moverCursorIzquierda(figura.tamano())
22        self.moverCursorArriba(figura.tamano())
23    }
24    if (figura.tipo() == "cuadrado") {
25        // en rojo
26        self.lineaAbajo(figura.tamano(), colorRojo)
27        self.lineaDerecha(figura.tamano(), colorRojo)
28        self.lineaArriba(figura.tamano(), colorRojo)
29        self.lineaIzquierda(figura.tamano(), colorRojo)
30        // en verde
31        self.lineaAbajo(figura.tamano(), colorVerde)
32        self.lineaDerecha(figura.tamano(), colorVerde)
33        self.lineaArriba(figura.tamano(), colorVerde)
34        self.lineaIzquierda(figura.tamano(), colorVerde)
35    }
36
37
38    if (figura.tipo() == "dobleLineaHorizontal") {
39        // en rojo
40        self.lineaDerecha(figura.tamano(), colorRojo)
41        self.moverCursorIzquierda(figura.tamano())
42        self.moverCursorArriba(1)
43        self.lineaDerecha(figura.tamano(), colorRojo)
44        self.moverCursorIzquierda(figura.tamano())
45        self.moverCursorAbajo(1)
46        // en verde
47        self.lineaDerecha(figura.tamano(), colorVerde)
48        self.moverCursorIzquierda(figura.tamano())
49        self.moverCursorArriba(1)
50        self.lineaDerecha(figura.tamano(), colorVerde)
51        self.moverCursorIzquierda(figura.tamano())
52        self.moverCursorAbajo(1)
53    }
54    self.apagarCursor()
55 }
56
57 method lineaAbajo(tamano, color) {
58     //implementacion que no interesa ...
59 }
60 method lineaDerecha(tamano, color) {
61     //implementacion que no interesa ...
62 }
63 method lineaArriba(tamano, color) {
64     //implementacion que no interesa ...
65 }
66 method lineaIzquierda(tamano, color) {
67     //implementacion que no interesa ...
68 }
69 method moverCursorDerecha(cuanto) {
70     //implementacion que no interesa ...
71 }
72
73 method moverCursorAbajo(cuanto) {
74     //implementacion que no interesa ...
75 }
76 method moverCursorIzquierda(cuanto) {
77     //implementacion que no interesa ...
78 }
79 method moverCursorArriba(cuanto) {
```

```

80     //implementacion que no interesa ...
81 }
82 method prenderCursor() {
83     //implementacion que no interesa ...
84 }
85 method apagarCursor() {
86     //implementacion que no interesa ...
87 }
88
89 }
90
91 object colorRojo {
92 }
93
94 object colorVerde {
95 }
96
97 class Figura {
98     var tamaño
99     var tipo
100
101     constructor(_tipo, _tamaño) {
102         tipo = _tipo
103         tamaño = _tamaño
104     }
105     method tamaño() { return tamaño }
106     method tipo() { return tipo }
107 }

```

Listing 22: polimorfismoPantalla.wlk

Ejercicio 13: Movimientos del ajedrez

Reorganice y modifique el siguiente código utilizando polimorfismo para mejorar la distribución de responsabilidades. Se permite (de hecho, se alienta a) agregar más clases a las planteadas, pasando parte del código a las nuevas clases.

El objetivo de este código es saber si una pieza de ajedrez puede moverse a una determinada posición. Las condiciones están simplificadas respecto de las reglas verdaderas del juego.

```

1  class TableroDeAjedrez {
2  method puedeMover(pieza, fila, columna) {
3      var puede
4      if (pieza.esPeon()) {
5          puede = (pieza.columna() == columna)
6                  and (pieza.fila() + 1 == fila)
7      }
8      if (pieza.esTorre()) {
9          puede = (pieza.columna() == columna)
10                 or (pieza.fila() == fila)
11      }
12     if (pieza.esRey()) {
13         puede = ((pieza.columna() - columna).abs() <= 1)
14                 and ((pieza.fila() - fila).abs() <= 1)
15     }
16     if (pieza.esAlfil()) {
17         puede = (pieza.columna() - columna).abs()
18                 == (pieza.fila() - fila).abs()
19     }
20     return puede
21         and not (pieza.estaEn(fila, columna))
22         and self.esPosicion(fila, columna)
23 }
24
25 method esPosicion(fila, columna) {

```

```

26     return fila.between(1,8) and columna.between(1,8)
27   }
28 }
29
30 class Pieza {
31   var fila
32   var columna
33   var tipo
34
35   method estaEn(f,c) {
36     return f == fila and c == columna
37   }
38
39   method fila() = fila
40   method columna() = columna
41
42   method esRey() { return tipo == 'Rey' }
43   method esTorre() { return tipo == 'Torre' }
44   method esAlfil() { return tipo == 'Alfil' }
45   method esPeon() { return tipo == 'Peon' }
46 }

```

Listing 23: polimorfismoAjedrez.wlk

Ejercicio 14: Ensamblador

Hace ya unos años que Volkswagen Argentina fabrica en su planta de Pacheco los modelos: Fox y Amarok (en sus dos versiones: volante derecho y volante izquierdo). Se está pensando fabricarlos en la misma planta el modelo Vento a la línea de ensamblado, con lo cual nos pidieron modificar el sistema.

Revisando el código, encontramos lo siguiente:

```

1  object lineaDeEnsamblado {
2    method ensamblar(unAuto, unColor) {
3      if (unAuto.modelo() == "Fox") {
4        unAuto.ponerPuertasDelanteras()
5        unAuto.ponerRuedas(4)
6        unAuto.ponerRuedaAuxilio()
7        unAuto.ponerPuertasTraseras()
8        unAuto.ponerVolanteAIzquierda()
9      }
10     if (unAuto.modelo() == "Amarok Izq") {
11       unAuto.ponerPuertasDelanteras()
12       unAuto.ponerRuedas(4)
13       unAuto.ponerRuedaAuxilio()
14       unAuto.ponerCajuela()
15       unAuto.ponerVolanteAIzquierda()
16     }
17     if (unAuto.modelo() == "Amarok Der") {
18       unAuto.ponerPuertasDelanteras()
19       unAuto.ponerRuedas(4)
20       unAuto.ponerRuedaAuxilio()
21       unAuto.ponerCajuela()
22       unAuto.ponerVolanteADerecha()
23     }
24     unAuto.pintarDe(unColor)
25   }
26 }
27
28 class Auto {
29   var modelo //Un string que indica el modelo del auto
30   var color //Un string que indica el color del auto
31   var cosas = #{ } /* todas las cosas (ruedas, cajuela, etc,) que se
32     * ponen son objetos que se agregan a la colección*/
33
34   constructor(unModelo) {
35     modelo = unModelo

```

```

36 }
37
38 method modelo() {
39     return modelo
40 }
41
42 method pintarDe(unColor) {
43     color = unColor
44 }
45
46 method color() {
47     return color
48 }
49 /*
50 * Todos los métodos poner son parecidos, se escribe la implementación
51 * de uno para tener una idea
52 */
53 method ponerPuertasDelanteras() {
54     cosas.add( new PuertasDelanteras() )
55 }
56 method ponerPuertasTraseras() {
57     ...
58 }
59 method ponerRuedaAuxilio () {
60     ...
61 }
62 method ponerRuedas(unaCantidad) {
63     ...
64 }
65 method ponerCajuela(unaCantidad) {
66     ...
67 }
68 method ponerVolanteAIzquierda() {
69     ...
70 }
71
72 method ponerVolanteADerecha() {
73     ...
74 }
75 }
76
77 test "construir un Amarok con Volante a la izquierda" {
78
79     var amarok = new Auto("Amarok Izq")
80     lineaDeEnsamblado.ensamblar(amarok, "azul")
81     //solo interesa la parte del test donde se muestra la creación
82     ...
83 }

```

Listing 24: polimorfismoAutos.wlk

Se pide:

1. Criticar la solución propuesta (Especial atención en la repetición de código, la distribución de responsabilidades y el uso o ausencia de polimorfismo).
2. Indicar qué se debe modificar en este código para agregar el modelo Vento. Escribir cómo sería la línea que construye un auto Vento, es decir, a qué clase se le hace *new* y en caso de recibir parámetro/s en el constructor, cuál/es es/son.
3. Refactorizar el código para solucionar los problemas detectados en a. ¡Vale agregar nuevas clases y mensajes!. Indicar cómo quedarían las primeras dos líneas del test.

Ejercicio 15: Centrales eléctricas

En un modelo del sistema de producción y transmisión eléctrica de un país, tenemos las clases `SistemaInterconectado`, `CentralHidrica`, `CentralAtomica` y `CentralACarbon`. Un sistema inteconectado incluye varias centrales. La capacidad de generación de un sistema inteconectado es la suma de la capacidad de generación de cada una de sus centrales. Para calcularla se agregó este método

```

1  class SistemaInterconectado {
2      var centrales = #{ }
3
4      method capacidadDeGeneracion() {
5          var total = 0
6          centrales.forEach({ central =>
7              if (central.esHidrica()) {
8                  total = total + central.cantTurbinas() *
9                      central.potenciaMaxTurbina() *
10                     central.caudalActual().max(central.caudadDeSaturacion()) /
11                     central.caudadDeSaturacion()
12              }
13              if (central.esAtomica()) {
14                  central.sectores().forEach({ sector =>
15                      total = total + sector.capacidadDeGeneracion() })
16              } if (central.esACarbon()) {
17                  total = total + central.capacidadInstalada()
18              }
19          return total
20      })
21  }
22 }
```

Listing 25: polimorfismoCentrales.wlk

Se pide:

- Suponiendo que las centrales hídricas son las únicas que responden `true` cuando se les pregunta `esHidrica()` y análogamente para las otras, indicar a través de un diagrama de clase qué mensajes entienden los distintos tipos de central según lo que se puede apreciar en el método `capacidadDeGeneracion()` de la clase `SistemaInterconectado`
- Para otro sistema que está haciendo el mismo grupo de desarrolladores se deben modelar plantas de aluminio. Una planta de aluminio tiene una única central generadora de energía (que puede ser hídrica, atómica o a carbón) de la cual se va a necesitar saber su capacidad de generación. Supongamos que las centrales solamente entienden los mensajes que se utilizan en el método `capacidadDeGeneracion()` de la clase `SistemaInterconectado`. Hay un problema de asignación de responsabilidades que impide usar el modelo de centrales eléctricas para la planta de aluminio. Indicar cuál es el problema (o sea, qué objeto está haciendo una cosa que no le corresponde) y cómo corregirlo (o sea, qué método/s habría que agregar o modificar en qué clase/s, escribiendo el código correspondiente).
- Qué concepto se está utilizando en su solución del punto b que no se estaba aprovechando en la solución propuesta por el enunciado?

Ejercicio 16: Comedero

- a) Analice el siguiente código y escriba un breve párrafo acerca de la distribución de responsabilidades entre los objetos/clases involucradas.
- b) Modifique el código utilizando polimorfismo para mejorar la distribución de responsabilidades.

```

1  object comedero {
2      var deposito = new Deposito()
3
4      method alimentar(unAnimal) {
5          if (unAnimal.esVaca()) {
6              deposito.descontarPasto(1)
7              unAnimal.darPasto(1)
8
9              deposito.descontarAlfalfa(2)
10             unAnimal.darAlfalfa(2)
11
12             deposito.descontarVitaminas(3)
13             unAnimal.darVitaminas(3)
14
15             deposito.descontarAgua(2)
16             unAnimal.darAgua(2)
17         }
18         if (unAnimal.esCaballo()) {
19             deposito.descontarPasto(1)
20             unAnimal.darPasto(1)
21
22             deposito.descontarAlfalfa(2)
23             unAnimal.darAlfalfa(2)
24
25             //los caballos necesitan vitaminas distintas
26             //con respecto a las vacas
27             deposito.descontarVitaminas(4)
28             unAnimal.darVitaminas(4)
29
30             deposito.descontarAgua(2)
31             unAnimal.darAgua(2)
32         }
33
34         if (unAnimal.esPerro()) {
35             deposito.descontarCarne(0.5)
36             unAnimal.darCarne(0.5)
37
38             deposito.descontarAgua(2)
39             unAnimal.darAgua(2)
40         }
41     }
42 }
43
44 class Animal {
45     var tipoDeAnimal
46     var aguaConsumida = 0
47     var energia = 100
48     var vitaminasConsumidas = 0
49
50     constructor(_tipo) {
51         tipoDeAnimal = _tipo
52     }
53     //Mensajes/Metodos para saber que animal es
54     //solo usados por el comedero
55     method esVaca() { return tipoDeAnimal == "vaca" }
56     method esCaballo() { return tipoDeAnimal == "caballo" }
57     method esPerro() { return tipoDeAnimal == "perro" }
58
59     //Mensaje/Metodo que interesa ser consultado

```

```
60 //para cualquier animal
61 method estaContento() {
62     return ( aguaConsumida * 10 / energia > 1 )
63     or ( aguaConsumida > 10 and
64         energia > 200 and vitaminasConsumidas > 0 )
65 }
66
67 /*metodos relacionados con la alimentacion */
68 method darAgua(litros) { aguaConsumida += litros }
69 method darVitaminas(cantidad) { vitaminasConsumidas += cantidad }
70
71 method darCarne(kilos) {
72     if (not self.esPerro()) {
73         error.throwWithMessage(
74             "Solo los carnivoros pueden consumir carne")
75     }
76     energia += kilos * 2
77 }
78
79 method darAlfalfa(kilos) {
80     if (not self.esCaballo() or self.esVaca()) {
81         error.throwWithMessage("
82             Solo el ganado pueden consumir alfalfa")
83     }
84     energia += kilos
85 }
86
87 method darPasto(kilos) {
88     if (not self.esCaballo() or self.esVaca()) {
89         error.throwWithMessage(
90             "Solo el ganado pueden consumir pasto")
91     }
92     energia += kilos * 0.1
93 }
94 }
95
96 //Desde aca para abajo asumir que es todo correcto
97 //Las clases Deposito y Alacena
98 //estan escritas para contextualizar el codigo anterior
99
100 class Deposito {
101     var litrosDisponiblesAgua = new Alacena(1000)
102     var kilosDisponiblesAlfalfa = new Alacena(500)
103     var kilosDisponiblesPasto = new Alacena(500)
104     var kilosDisponiblesCarne = new Alacena(100)
105     var unidadesDisponiblesVitaminas = new Alacena(200)
106
107     method descontarPasto(kilos) { kilosDisponiblesPasto.decrementar(kilos) }
108
109     method descontarAlfalfa(kilos) {
110         kilosDisponiblesAlfalfa.decrementar(kilos)
111     }
112
113     method descontarAgua(litros) { litrosDisponiblesAgua.decrementar(litros) }
114     method descontarCarne(kilos) { kilosDisponiblesCarne.decrementar(kilos) }
115     method descontarVitaminas(unidades) {
116         unidadesDisponiblesVitaminas.decrementar(unidades)
117     }
118 }
119 class Alacena {
120     var disponible
121
122     constructor(_disponible) {
123         disponible = _disponible
124     }
125
126     method decrementar(unaCantidad) {
127         if (unaCantidad > disponible) {
128             error.throwWithMessage(
129                 "No hay " + unaCantidad +
130                 " disponible. Maximo: " +
```

```

131     disponible)
132   }
133   disponible -= unaCantidad
134 }
135 }

```

Listing 26: polimorfismoComedero.wlk

Ejercicio 17: Misiones

Modifique el código utilizando polimorfismo, organización del comportamiento entre las clases, y (en un caso) mensajes adecuados a las colecciones, para mejorar la distribución de responsabilidades. Evite el código duplicado.

```

1  class Mision {
2
3  var objetivo //Es una instancia de la clase
4      //Objetivo u ObjetivoMultiple)
5  var mundo // Es una instancia de la clase Mundo
6  constructor(_objetivo, _mundo) {
7      objetivo = _objetivo
8      mundo = _mundo
9  }
10
11  method esExitosa() {
12      if(objetivo.esSupervivencia()) {
13          return not mundo.destruido()
14              and mundo.tiempoActual() >= objetivo.tiempoDeSupervivencia()
15              and mundo.personajeEstaVivo()
16      }
17      if(objetivo.esParaDerrotarEnemigos()) {
18          return not mundo.destruido()
19              and mundo.enemigosDerrotados() >= objetivo.enemigosADerrotar()
20      }
21      if(objetivo.esParaDefenderAliados()) {
22          return not mundo.destruido()
23              and mundo.aliadosCaidos() <= objetivo.aliadosPrescindibles()
24      }
25      if(objetivo.esMultiple()) {
26          return self.resolverExitoMultiple(objetivo)
27      }
28      error.throwWithMessage("Tipo de objetivo desconocido")
29      return false
30  }
31  //metodo auxiliar que devuelve true
32  //si se cumplieron TODOS los objetivos del objetivoMultiple
33  method resolverExitoMultiple(objetivoMultiple) {
34      var valorRetorno = true
35      objetivoMultiple.objetivosInternos().forEach({unObjetivo =>
36          if(unObjetivo.esSupervivencia()) {
37              valorRetorno = valorRetorno
38                  and not mundo.destruido()
39                  and mundo.tiempoActual() >= objetivo.tiempoDeSupervivencia()
40                  and mundo.personajeEstaVivo()
41          }
42          if(unObjetivo.esParaDerrotarEnemigos()) {
43              valorRetorno = valorRetorno
44                  and not mundo.destruido()
45                  and mundo.enemigosDerrotados() >= objetivo.enemigosADerrotar()
46          }
47          if(unObjetivo.esParaDefenderAliados()) {
48              valorRetorno = valorRetorno
49                  and not mundo.destruido()
50                  and mundo.aliadosCaidos() <= objetivo.aliadosPrescindibles()
51          }
52          if(unObjetivo.esMultiple()) {
53              valorRetorno = valorRetorno
54                  and self.resolverExitoMultiple(unObjetivo)
55              //Si tiene un objetivo multiple dentro de otro llama

```

```

56         //al mismo metodo pero con el interno esto funciona
57         //(termina la ejecucion con el resultado esperado)
58     }
59
60     })
61     return valorRetorno
62 }
63 }
64 class Objetivo {
65
66     //tiempo que si se supera la mision es exitosa
67     var tiempoDeSupervivencia
68
69     //cantidad de enemigos que se deben derrotar para el exito
70     var enemigosADerrotar
71
72     //cantidad de aliados maxima que se pueden perder
73     //Si se supera se fracasa
74     var aliadosPrescindibles
75
76     //*****Metodos para configurar el objetivo*****
77
78     method configurarComoSupervivencia(_tiempo) {
79         tiempoDeSupervivencia = _tiempo
80         enemigosADerrotar = null
81         aliadosPrescindibles = null
82     }
83
84     method configurarParaDerrotarEnemigos(_enemigosADerrotar) {
85         tiempoDeSupervivencia = null
86         enemigosADerrotar = _enemigosADerrotar
87         aliadosPrescindibles = null
88     }
89
90     method configurarParaDefenderAliados(_prescindibles) {
91         tiempoDeSupervivencia = null
92         enemigosADerrotar = null
93         aliadosPrescindibles = _prescindibles
94     }
95
96     //*****Metodos para saber el tipo de Objetivo*****
97     method esSupervivencia() {return tiempoDeSupervivencia != null}
98     method esParaDerrotarEnemigos() {return enemigosADerrotar != null}
99     method esParaDefenderAliados() {return aliadosPrescindibles != null}
100    method esMultiple() {return false}
101
102    //*****Getters para exponer el estado interno *****
103
104    method tiempoDeSupervivencia() {return tiempoDeSupervivencia}
105    method enemigosADerrotar() {return enemigosADerrotar}
106    method aliadosPrescindibles() {return aliadosPrescindibles}
107 }
108
109 class ObjetivoMultiple {
110     //Es una coleccion de objetivos, instancias de las clases
111     //Objetivo y/o ObjetivoMultiple
112     var objetivosInternos
113
114     constructor (_objetivosInternos) {objetivosInternos = _objetivosInternos}
115     //*****Metodos para saber el tipo de Objetivo*****
116     method esSupervivencia() {return false}
117     method esParaDerrotarEnemigos() {return false}
118     method esParaDefenderAliados() {return false}
119     method esMultiple() {return true}
120     //*****Getters para exponer el estado interno *****
121     method objetivosInternos(){return objetivosInternos}
122 }
123
124
125 //La implementacion de esta clase no interesa a los efectos del examen
126 //Solo se muestran los mensajes que entiende

```

```

127 class Mundo {
128     method tiempoActual() //Devuelve el tiempo actual
129     method enemigosDerrotados() // Devuelve la cantidad de enemigos derrotados
130     method aliadosCaidos() //Devuelve la cantidad de aliados caidos
131     method destruido() //Devuelve true si se destruyo el mundo
132     method personajeEstaVivo() //Devuelve si el personaje principal esta Vivo
133 }

```

Listing 27: polimorfismoMisiones.wlk

```

1
2 test "ambiente" {
3
4     var mundo = new Mundo()
5     var obj1 = new Objetivo()
6     obj1.configurarComoSupervivencia(1500)
7     var obj2 = new Objetivo()
8     obj2.configurarParaDefenderAliados(5)
9     var objetivo = new ObjetivoMultiple(#{obj1, obj2})
10    var mision = new Mision(objetivo, mundo)
11    assert.notThat(mision.esExitosa())
12
13 }

```

Listing 28: polimorfismoMisiones.wtest

Ejercicio 18: Turismo

Modifique el código utilizando polimorfismo, organización del comportamiento entre las clases, y (en un caso) mensajes adecuados a las colecciones, para mejorar la distribución de responsabilidades. Evite el código duplicado. Mejore si corresponde los lanzamientos de errores.

```

1 class Paquete {
2
3     var cantidadPersonas //un numero
4     var servicios //coleccion de instancias de la clase Servicio
5     var premium //un booleano
6     var reservado = false
7
8     constructor(_servicios, _cantidadPersonas, _premium) {
9         cantidadPersonas = _cantidadPersonas
10        servicios = _servicios
11        premium = _premium
12    }
13
14    method estaReservado() {return reservado}
15
16    //Se puede reservar si no esta reservado y
17    //TODOS los servicios se pueden reservar
18    method sePuedeReservar() {
19
20        if(self.estaReservado()) {
21            return false
22        }
23        var todosSePuedenReservar = true
24        servicios.forEach({unServicio =>
25
26            if(unServicio.esHotel()) {
27                //Un hotel se puede reservar si hay lugares disponibles.
28                //Ademas, si el paquete es premium el hotel tiene que ser
29                //como minimo de 4 estrellas
30                todosSePuedenReservar = todosSePuedenReservar and
31                unServicio.lugaresDisponibles() >= cantidadPersonas and
32                (not premium or unServicio.estrellas() >= 4)
33            }
34            if(unServicio.esVehiculoParaTralado()) {

```

```

35     //Un traslado se puede reservar si el vehiculo cuenta con
36     //lugares disponibles y tiene la verificacion tecnica al dia.
37     //Si el paquete es premium, tambien tiene que cumplir que
38     //tenga aire Acondicionado
39     todosSePuedenReservar = todosSePuedenReservar and
40     unServicio.lugaresDisponibles() >= cantidadPersonas and
41     unServicio.tieneVTV() and
42     (not premium or unServicio.tieneAireAcondicionado())
43 }
44 if(unServicio.esTour()) {
45     //Un tour tiene un vehiculo para traslado y opcionalmente
46     // seguridad privada
47     //Para que se pueda reservar se tienen que cumplir todas las
48     //condiciones sobre el vehiculo (que son las mismas que se
49     //piden a los servicios que son vehiculos para traslados)
50     // Ademas, el tour tiene que tener espacio disponible que es
51     //independiente de los lugares del vehiculo. (Por ejemplo,
52     //se puede hacer un tour para 10 personas en una combi para 15).
53     // Si es premium, el tour tiene que tener seguridad privada
54
55     todosSePuedenReservar = todosSePuedenReservar and
56     unServicio.lugaresDisponibles() >= cantidadPersonas and
57     unServicio.vehiculoTrasladoDeTour().lugaresDisponibles()
58     >= cantidadPersonas and
59     unServicio.vehiculoTrasladoDeTour().tieneVTV() and
60     (not premium or
61     (unServicio.vehiculoTrasladoDeTour().tieneAireAcondicionado()
62     and unServicio.tieneSeguridadPrivada()))
63 }
64 })
65 return todosSePuedenReservar
66 }
67
68 //Cuando se reserva se cambia el estado y se modifican los lugares
69 //disponibles en los servicios
70 // IMPORTANTE: Analizar que sucede si se pide reservar a un paquete
71 // que no cumple con las condiciones.
72 //Modificar en caso de que la estrategia fuera incorrecta
73 method reservar() {
74     reservado = true
75     servicios.forEach( {unServicio =>
76         unServicio.lugaresDisponibles(
77         unServicio.lugaresDisponibles()-cantidadPersonas)
78         if(unServicio.esTour()) {
79             unServicio.vehiculoTrasladoDeTour().lugaresDisponibles(
80             unServicio.vehiculoTrasladoDeTour().lugaresDisponibles()
81             - cantidadPersonas)
82         }
83     })
84 }
85
86 }
87
88 class Servicio {
89
90     var lugaresDisponibles //lugares disponibles del SERVICIO
91     var estrellas = 1//cantidad de estrellas del HOTEL
92     var aireAcondicionado = false //si el VEHICULO tiene aire acond.
93     var vtv = true //si el VEHICULO tiene la vtv
94     var seguridadPrivada = false //si el TOUR tiene seguridad privada
95     var vehiculoParaTrasladoDeTour = null // es otra instancia de
96         //Servicio que tiene la
97         //informacion del vehiculo
98         //que usa el TOUR
99
100     var esHotel = false
101     var esTour = false
102     var esVehiculoParaTraslado = false
103
104     constructor(_lugaresDisponibles) {
105         lugaresDisponibles= _lugaresDisponibles

```

```

106     }
107     method configurarComoHotel(_estrellas) {
108         esHotel = true
109         esTour = false
110         esVehiculoParaTraslado = false
111         estrellas = _estrellas
112     }
113     method configurarComoVehiculo(_tieneAire, _tieneVtv) {
114         esHotel = false
115         esTour = false
116         esVehiculoParaTraslado = true
117         aireAcondicionado = _tieneAire
118         vtv = _tieneVtv
119     }
120     method configurarComoTour(_vehiculoTraslado, _seguridadPrivada) {
121         esHotel = false
122         esTour = true
123         esVehiculoParaTraslado = false
124         vehiculoParaTrasladoDeTour = _vehiculoTraslado
125         seguridadPrivada = _seguridadPrivada
126     }
127
128     method lugaresDisponibles() {return lugaresDisponibles}
129     method lugaresDisponibles(_lugaresDisponibles) {
130         lugaresDisponibles = _lugaresDisponibles
131     }
132     method esHotel() {return esHotel}
133     method esTour() {return esTour}
134     method esVehiculoParaTraslado() {return esVehiculoParaTraslado}
135     method estrellas() {return estrellas}
136     method tieneAireAcondicionado() {return aireAcondicionado}
137     method tieneVTV() {return vtv}
138     method vehiculoTrasladoDeTour() {return vehiculoParaTrasladoDeTour}
139     method tieneSeguridadPrivada() {return seguridadPrivada}
140 }

```

Listing 29: polimorfismoTurismo.wlk

```

1 test "reservaExitosa" {
2     var luchoHotel = new Servicio(50)
3     var micro = new Servicio(30)
4     var combi = new Servicio(15)
5     var cityTour = new Servicio(10)
6     luchoHotel.configurarComoHotel(2)
7     micro.configurarComoVehiculo(true, true)
8     combi.configurarComoVehiculo(true, true)
9     cityTour.configurarComoTour(combi, false)
10    var paquete = new Paquete(#{luchoHotel, micro, cityTour}, 2, false)
11
12    assert.that(paquete.sePuedeReservar())
13    paquete.reservar()
14    assert.that(paquete.estaReservado())
15    assert.equals(48, luchoHotel.lugaresDisponibles())
16    assert.equals(28, micro.lugaresDisponibles())
17    assert.equals(8, cityTour.lugaresDisponibles())
18    assert.equals(13, combi.lugaresDisponibles())
19 }

```

Listing 30: polimorfismoTurismo.wtest

Ejercicio 19: Transporte

Modifique el código utilizando polimorfismo, organización del comportamiento entre las clases, y (en un caso) mensajes adecuados a las colecciones, para mejorar la distribución de responsabilidades.

```

1
2 class Viaje {

```

```

3
4   var medio //es una instancia de la clase MedioTransporte
5   var origen //es una ubicacion
6   var destino //es una ubicacion
7
8   constructor(_medio, _origen, _destino) {
9       medio = _medio
10      origen = _origen
11      destino = _destino
12  }
13
14  /**Devuelve true si el viaje puede ser realizado */
15  method esPosible() {
16
17      if(medio.esCaminante()) {
18          return self.hayCamino(origen, destino)
19              and self.distancia(origen, destino) < medio.distanciaMaxima()
20      }
21      if(medio.esAuto()) {
22          return self.hayCalles(origen, destino)
23              and self.distancia(origen, destino) < medio.distanciaMaxima()
24      }
25      if(medio.esColectivo()) {
26          var paradaSubir = medio.paradaCercana(origen)
27          var paradaBajar = medio.paradaCercana(destino)
28          var caminante = new MedioTransporte("caminante", 800)
29          var caminoHastaSubida = new Viaje(caminante, origen, paradaSubir)
30          var caminoDesdeBajada = new Viaje(caminante, paradaBajar, destino)
31          return caminoHastaSubida.esPosible()
32              and caminoDesdeBajada.esPosible()
33              and self.distancia(origen, destino) < medio.ditanciaMaxima()
34      }
35
36      error.throwWithMessage("No se reconoce el tipo del medio")
37  }
38
39  //Este es un metodo auxiliar usado solo en esPosible()
40  method distancia(desde, hasta) {
41      if(medio.esCaminante()) {
42          return navegador.distanciaPeatonal(desde, hasta)
43      }
44      if(medio.esAuto()) {
45          return navegador.distanciaPorCaminoTransitable(desde, hasta)
46      }
47      if(medio.esColectivo()) {
48          var tramos = medio.tramosEntre(desde, hasta)
49          var sumatoria = 0
50          tramos.foreach({unTramo =>
51              sumatoria = sumatoria
52                  + navegador.distanciaPorCaminoTransitable(
53                      unTramo.origen(), unTramo.destino())
54          })
55          return sumatoria
56      }
57
58      error.throwWithMessage("No se reconoce el tipo del medio")
59  }
60
61  //Este es un metodo auxiliar usado solo en esPosible()
62  //para los autos
63  method hayCalles(orig, dest) {
64      return navegador.existeCaminoTransitable(orig, dest)
65  }
66
67  //Este es un metodo auxiliar usado solo en esPosible(),
68  //para los caminantes
69  method hayCamino(orig, dest) {
70      return navegador.existeCaminoPeatonal(orig, dest)
71  }
72 }
73

```

```
74 class MedioTransporte {
75     var tipo
76     var distanciaMaxima
77     var tramos
78
79     //constructor para autos y caminantes
80     //por eso no interesa inicializar la variable
81     //tramos
82     constructor (_tipo, _distanciaMaxima){
83         tipo = _tipo
84         distanciaMaxima = _distanciaMaxima
85     }
86
87     //si es un colectivo hay que usar este constructor
88     //porque si no fallan los metodos tramosEntre y paradaCercana
89     constructor (_tipo, _distanciaMaxima, _tramos){
90         tipo = _tipo
91         distanciaMaxima = _distanciaMaxima
92         tramos = _tramos
93     }
94
95     method esCaminante() {return tipo == "caminante"}
96     method esAuto() {return tipo == "auto"}
97     method esColectivo() {return tipo == "colectivo"}
98     method distanciaMaxima() {return distanciaMaxima}
99
100     /**
101     * Para colectivos: devuelve todos los tramos del recorrido
102     * que hay entre la parada mas cercana al origen
103     * y la parada mas cercana al destino
104     */
105     method tramosEntre(origen, destino) {
106         //ATENCIÓN: Esta implementacion usa la variable de
107         //instancia tramos y el metodo self.paradaCercana(ubicacion)
108     }
109     /**
110     * Para colectivos: devuelve una ubicacion que esta dentro
111     * de un tramo del recorrido
112     */
113     method paradaCercana(ubicacion) {
114         //ATENCIÓN: Esta implementacion usa la variable de instancia tramos
115     }
116 }
117 /**
118 * Modela una ubicacion en el mundo.
119 * Todas las referencias a un origen, un destino,
120 * un lugar desde o un lugar hasta
121 * (en Viaje, TramoDeColectivo y navegador),
122 * y las paradas de colectivo, son instancias de esta clase
123 */
124 class Ubicacion {
125     //implementacion
126 }
127 /**
128 * Modela un tramo de un recorrido de un colectivo
129 * El recorrido de un colectivo entre dos paradas
130 * es una lista de instancias de esta clase
131 */
132 class TramoDeColectivo {
133     method origen() {/*implementacion*/}
134     method destino() {/*implementacion*/}
135 }
136 object navegador {
137     /**Devuelve si existe al menos un camino
138     * usando caminos peatonales y/o calles */
139     method existeCaminoPeatonal(orig, dest) {/*implementacion*/}
140
141     /**Devuelve la distancia entre dos puntos
142     * usando caminos peatonales y/o calles */
143     method distanciaPeatonal(orig, dest) {/*implementacion*/}
144 }
```

```
145  /**Devuelve la distancia entre dos puntos usando calles */
146  method distanciaPorCaminoTransitable(orig, dest) {/*implementacion*/}
147
148  /**Devuelve si existe al menos un camino usando calles */
149  method existeCaminoTransitable(orig, dest) {/*implementacion*/}
150 }
```

Listing 31: polimorfismoTransporte.wlk

3. Referencias

Ejercicio 20: Mascotas

Teniendo en cuenta las siguientes clases:

```

1 class Animal {
2     var duenio
3     method duenio() = duenio
4     method duenio(_duenio) { duenio = _duenio }
5     method aguante() { return 3 }
6 }
7 class Perro inherits Animal {
8     override method aguante() { return super() + 2 }
9 }
10
11 class Duenio { }
```

Listing 32: referenciasMascotas.wlk

Parte 1: Estado inicial

Construya el grafo de objetos luego de ejecutar el siguiente código:

```

1 var hijitus = new Duenio()
2 var linyera = new Duenio()
3 var tom = new Duenio()
4 var pichichus = new Perro()
5 pichichus.duenio(hijitus)
6 var diogenes = new Perro()
7 diogenes.duenio(linyera)
8 var garfield = new Animal()
9 garfield.duenio(tom)
10 var animales = #{pichichus, diogenes, garfield}
```

Listing 33: referenciasMascotas.wpgm

Parte 2: Estado Final

Tomando el estado configurado como inicial, ejecute el siguiente código y construya el grafo de objetos final. En caso de que alguna línea lance un error, indique el motivo del mismo e ignórela.

```

1 pichichus.duenio(garfield.duenio())
2 hijitus = garfield.aguante()
3 diogenes = diogenes.aguante()
4 animales.remove(diogenes)
5 pichichus = hijitus == animales.size()
6 animales.add(garfield.duenio())
7 var duenios = animales.map({x => x.duenio()})
```

Listing 34: referenciasMascotas.wpgm

Ejercicio 21: Empresas

Teniendo en cuenta las siguientes clases:

```

1 class Persona {}
2 class Empresa {
3     var director
4     method director() = director
5     method director(_director) {
6         director = _director
7     }
8 }
```

Listing 35: referenciasEmpresas.wlk

Parte 1: Estado inicial

Construya el grafo de objetos luego de ejecutar el siguiente código:

```

1 var mexicano = new Persona()
2 var chino = new Persona()
3 var argentino = new Persona()
4 var saraza = new Empresa()
5 var cadornaCo = new Empresa()
6 saraza.director(chino)
7 cadornaCo.director(argentino)
8 var empresas = [saraza, cadornaCo]
9 var directores

```

Listing 36: referenciasEmpresas.wpgm

Parte 2: Estado Final

Tomando el estado configurado como inicial, ejecute el siguiente código y construya el grafo de objetos final. En caso de que alguna línea lance un error, indique el motivo del mismo e ignórela.

```

1 mexicano = cadornaCo
2 chino.director(chino)
3 cadornaCo.director(mexicano)
4 empresas.add(chino)
5 directores = empresas.map({unaEmpresa => unaEmpresa.director()})
6 directores = empresas
7 directores.remove(chino)
8 directores = empresas.map({unaEmpresa => unaEmpresa.director()})

```

Listing 37: referenciasEmpresas.wpgm

Ejercicio 22: Sobrinos de Donald

Teniendo en cuenta las siguientes clases:

```

1 class Sociable {
2   var amigo
3   method amigo(_amigo) { amigo = _amigo }
4   method amigo() { return amigo }
5 }
6
7 class Ermitanio {
8   method amigo() { return self }
9 }
10
11 class Casa {
12   var personas = [] //es una LISTA!! (no conjunto)
13
14   method agregarPersona(_persona) { personas.add(_persona) }
15   method personas() { return personas }
16   method masNuevo() { return personas.last() }
17 }
18
19 class Club inherits Casa {
20   override method agregarPersona(_persona) {
21     super(_persona.amigo())
22   }
23 }

```

Listing 38: referenciasSobrinosDonald.wlk

Parte 1: Estado inicial

Construya el grafo de objetos luego de ejecutar el siguiente código:

```

1  var hugo = new Sociable()
2  var paco = new Sociable()
3  var luis = new Ermitanio()
4  hugo.amigo(paco)
5  paco.amigo(luis)
6  var hogar = new Casa()
7  hogar.agregarPersona(hugo)
8  hogar.agregarPersona(paco)
9  hogar.agregarPersona(luis)
10 var cortapalos = new Club()

```

Listing 39: referenciasSobrinosDonald.wpgm

Parte 2: Estado Final

Tomando el estado configurado como inicial, ejecute el siguiente código y construya el grafo de objetos final. En caso de que alguna línea lance un error, indique el motivo del mismo e ignórela.

```

1  cortapalos.agregarPersona(hugo)
2  cortapalos.agregarPersona(luis)
3  hugo.amigo(hogar.masNuevo())
4  paco = cortapalos
5  paco.agregarPersona(hogar)
6  var amigos = paco.personas().map({x => x.amigo()})
7  hugo.amigo(hugo)
8  luis.amigo(hugo)
9  var amigoDeHugo = hugo.amigo()

```

Listing 40: referenciasSobrinosDonald.wpgm

Ejercicio 23: Guitarras

Teniendo en cuenta las siguientes clases:

```

1
2  class Guitarra {
3    var cuerdas = 6
4    method cuerdas() = cuerdas
5    method cuerdas(_cuerdas) { cuerdas = _cuerdas }
6    method volumen() { return cuerdas * 10 }
7
8  }
9  class GuitarraElectrica inherits Guitarra{
10   var equipo
11   method equipo() { return equipo }
12   method equipo(_equipo) { equipo = _equipo }
13   override method volumen() {
14     return super() * equipo.ganancia()
15   }
16 }
17
18 class Equipo {
19   var ganancia = 10
20   method ganancia() = ganancia
21   method ganancia(_ganancia) { ganancia = _ganancia }
22 }

```

Listing 41: referenciasGuitarras.wlk

Parte 1: Estado inicial

Construya el grafo de objetos luego de ejecutar el siguiente código:

```

1  var criolla = new Guitarra()
2  var acustica12 = new Guitarra()
3  acustica12.cuerdas(12)
4  var lucille = new GuitarraElectrica()
5  var marshall = new Equipo()
6  lucille.equipo(marshall)

```

Listing 42: referenciasGuitarras.wpgm

Parte 2: Estado Final

Tomando el estado configurado como inicial, ejecute el siguiente código y construya el grafo de objetos final. En caso de que alguna línea lance un error, indique el motivo del mismo e ignórela.

```

1  marshall.ganancia(lucille.volumen())
2  acustica12.cuerdas(marshall.ganancia() / 100 )
3  var x = criolla.cuerdas() == acustica12.cuerdas()
4  lucille.equipo(criolla.cuerdas())
5  lucille.cuerdas(lucille.volumen())
6  lucille.equipo(marshall)
7  criolla = marshall
8  marshall = new Equipo()
9  marshall.ganancia(5)
10 var volumenFinal = lucille.volumen()

```

Listing 43: referenciasGuitarras.wpgm

Ejercicio 24: Bonificaciones

Dadas las siguientes clases:

```

1  class Empleado {
2    var bonificacion = 0
3
4    method bonificar(unNumero) { bonificacion = unNumero }
5    method bonificacion() { return bonificacion }
6    method preferido() { return self }
7  }
8
9  class Jefe inherits Empleado {
10
11    var subordinados = #{}
12
13    method addEmpleado (unEmpleado) { subordinados.add(unEmpleado) }
14
15    override method bonificar(unNumero) {
16      bonificacion = bonificacion.max(unNumero)
17      subordinados.forEach(
18        {subordinado => subordinado.bonificar(unNumero)}
19      )
20    }
21
22    override method preferido() {
23      return subordinados.min(
24        {subordinado => subordinado.bonificacion()}
25      )
26    }
27  }
28 }

```

Listing 44: referenciasBonificaciones.wlk

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  var juan = new Empleado()
2  var pedro = new Empleado()
3  var santiago = new Empleado()
4  var roque = new Jefe()
5  var luis = new Jefe()
6  roque.addEmpleado(juan)
7  roque.addEmpleado(pedro)
8  luis.addEmpleado(santiago)
9  luis.addEmpleado(roque)

```

Listing 45: referenciasBonificaciones.wpgm

Parte 1: Estado final

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  luis.bonificar(10)
2  roque.bonificar(2)
3  pedro.bonificar(5)
4  luis.addEmpleado(roque.preferido())
5  luis.addEmpleado(pedro.preferido())
6  var preferidoLuis = luis.preferido()

```

Listing 46: referenciasBonificaciones.wpgm

Ejercicio 25: Juguetes

Dada las siguientes clases:

```

1  class Juguete {
2    var color
3    constructor(_color) {color = _color}
4    method color(){return color}
5    method color(_color){color = _color}
6  }
7  object pelota {
8    method color() {return "rojo"}
9    //a la pelota no se le puede cambiar el color
10 }
11 class GrupoDeJuguetes {
12   var juguetes
13   constructor(_juguetes) {juguetes = _juguetes}
14   method agregarJuguete(juguete) {juguetes.add(juguete)}
15   method juguetes() {return juguetes}
16   method colores() {return juguetes.map({j => j.color()}).asSet()}
17   method pintar(color) {juguetes.forEach({j => j.color(color)}}}
18 }

```

Listing 47: referenciasJuguetes.wlk

Parte 1: Estado inicial

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  var pepa = new Juguete("rosa")
2  var mcqueen = new Juguete("rojo")
3  var minion = new Juguete("amarillo")
4  var patito = new Juguete("amarillo")
5  var enCasa = new GrupoDeJuguetes({pepa, mcqueen, minion, patito, pelota})
6  var paraViajar = new GrupoDeJuguetes({minion, patito})

```

Listing 48: referenciasJuguetes.wtest

Parte 2: Estado final

El siguiente código se ejecuta *luego del anterior*.

```

1  var x = patito == minion
2  pepa = pelota
3  var y = enCasa.colores().contains("rosa")
4  paraViajar.pintar("rojo")
5  var z = enCasa.colores().contains("amarillo")
6  mcqueen.color(patito)
7  var coloresDeCasa = enCasa.colores()
8  enCasa.pintar("amarillo")
9  paraViajar.colores().agregarJuguete(pelota)

```

Listing 49: referenciasJuguetes.wtest

Analizar para cada línea si se puede evaluar correctamente o si, por el contrario, se produce un error. Indique las líneas que producen errores y **las causas** de los mismos. Dibuje el grafo de objetos resultante de la evaluación, ignorando las líneas que producen errores. Recordar que si un objeto no entiende el mensaje se produce error. Así que sean cuidadosos porque puede pasar de manera intencional.

Ejercicio 26: Popeye

Dada las siguientes clases:

```

1  object popeye {
2    var comio = false
3    method comerEspinaca() {comio = true}
4    method fuerza() {return if(comio) 1000 else 5 }
5  }
6  class Marinero {
7    var fuerza = 10
8    method fuerza() {return fuerza}
9    method comerEspinaca() {fuerza = fuerza + 10}
10 }
11 class Barco {
12   var marineros = #{}
13   method agregarMarinero(marinero) {marineros.add(marinero)}
14   method marineros() {return marineros}
15   method alimentarMarineros() {
16     return marineros.forEach({m=>m.comerEspinaca()})
17   }
18   method masFuerte() {return marineros.max({m=>m.fuerza()})}
19 }

```

Listing 50: referenciasPopeye.wlk

Parte 1: Estado inicial

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  var brutus = new Marinero()
2  var olivia = new Marinero()
3  var cocoliso = new Marinero()
4  var perlaNegra = new Barco()
5  var interceptor = new Barco()
6  perlaNegra.agregarMarinero(brutus)
7  perlaNegra.agregarMarinero(olivia)
8  perlaNegra.agregarMarinero(popeye)

```

Listing 51: referenciasPopeye.wtest

Parte 2: Estado final

El siguiente código se ejecuta *luego del anterior*.

```

1  var x = brutus == olivia
2  brutus.comerEspinaca()
3  brutus = cocoliso
4  var y = perlaNegra.masFuerte()
5  var z = brutus == y
6  perlaNegra.alimentarMarineros()
7  interceptor.agregarMarinero(perlaNegra.masFuerte())
8  interceptor.agregarMarinero(perlaNegra)
9  interceptor.alimentarMarineros()

```

Listing 52: referenciasPopeye.wtest

Analizar para cada línea si se puede evaluar correctamente o si, por el contrario, se produce un error. Indique las líneas que producen errores y **las causas** de los mismos. Dibuje el grafo de objetos resultante de la evaluación, ignorando las líneas que producen errores. Recordar que si un objeto no entiende el mensaje se produce error. Así que sean cuidadosos porque puede pasar de manera intencional.

Ejercicio 27: Heman

Dada las siguientes clases:

```

1  object heman {
2    var transformado = false
3    method gritoDeGuerra() {transformado = true}
4    method fuerza() {return if(transformado) 2000 else 10 }
5  }
6  class MasterOfUniverse {
7    var fuerza = 20
8    method fuerza() {return fuerza}
9    method gritoDeGuerra() {fuerza = fuerza + 10}
10 }
11 class Castillo {
12   var maestros = #{}
13   method agregarMaestro(maestro) {maestros.add(maestro)}
14   method maestros() {return maestros}
15   method enfurecer() {
16     return maestros.forEach({m=>m.gritoDeGuerra()})
17   }
18   method masFuerte() {return maestros.max({m=>m.fuerza()})}
19 }

```

Listing 53: referenciasHeman.wlk

Parte 1: Estado inicial

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  var manatarms = new MasterOfUniverse()
2  var teela = new MasterOfUniverse()
3  var orko = new MasterOfUniverse()
4  var palacio = new Castillo()
5  var grayskull = new Castillo()
6  palacio.agregarMaestro(manatarms)
7  palacio.agregarMaestro(teela)
8  palacio.agregarMaestro(heman)

```

Listing 54: referenciasHeman.wtest

Parte 2: Estado final

El siguiente código se ejecuta *luego del anterior*.

```

1  var x = manatarms == teela
2  manatarms.gritoDeGuerra()
3  manatarms = orko
4  var y = palacio.masFuerte()
5  var z = manatarms == y
6  palacio.enfurecer()
7  grayskull.agregarMaestro(palacio.masFuerte())
8  grayskull.agregarMaestro(palacio)
9  grayskull.enfurecer()

```

Listing 55: referenciasHeman.wtest

Analizar para cada línea si se puede evaluar correctamente o si, por el contrario, se produce un error. Indique las líneas que producen errores y **las causas** de los mismos. Dibuje el grafo de objetos resultante de la evaluación, ignorando las líneas que producen errores. Recordar que si un objeto no entiende el mensaje se produce error. Así que sean cuidadosos porque puede pasar de manera intencional.

Ejercicio 28: Jugadores

Dada las siguientes clases:

```

1  class Jugador {
2    var idolo
3    var goles
4
5    constructor (_idolo, _goles) {
6      idolo = _idolo
7      goles = _goles
8    }
9
10   method idolo() {return idolo }
11
12   method idolo(_idolo) {idolo = _idolo}
13
14   method goles() {return goles}
15 }
16
17 object distefano {
18   method idolo() {return self}
19   method goles() {return 694}
20 }
21
22 class Equipo {
23   var jugadores
24   constructor(_jugadores){jugadores = _jugadores}
25   method idolos() { return new Equipo(jugadores.map({j=>j.idolo()}))}
26   method jugadores() {return jugadores}
27   method goleador() {
28     return jugadores.max({unJugador => unJugador.goles()})
29   }
30   method agregarJugador(jug) { jugadores.add(jug) }
31 }

```

Listing 56: referenciasJugadores.wlk

Parte 1: Estado inicial

Dibuje el grafo de objetos luego de ejecutar el siguiente código:

```

1  var bochini = new Jugador(distefano, 97)
2  var maradona = new Jugador(bochini, 352)

```

```
3 var riquelme = new Jugador(maradona, 165)
4 var francescoli = new Jugador(distefano, 244)
5 var zidane = new Jugador(francescoli, 153)
6 var estrellas = new Equipo([riquelme, zidane, maradona])
```

Listing 57: referenciasJugadores.wtest

Parte 2: Estado final

El siguiente código se ejecuta *luego del anterior*.

```
1 var historicos = estrellas.idolos()
2 var goleador = historicos.goleador()
3 historicos = historicos.idolos()
4 var megaGoleador = historicos.goleador()
5 bochini.idolo(maradona.idolo())
6 var goles = estrellas.sum({unJugador => unJugador.goles()})
7 historicos.agregarJugador(estrellas)
8 goleador = historicos.goleador()
```

Listing 58: referenciasJugadores.wtest

Analizar para cada línea si se puede evaluar correctamente o si, por el contrario, se produce un error. Indique las líneas que producen errores y cual es el error. Dibuje el grafo de objetos resultante de la evaluación, ignorando las líneas que producen errores.

Recordar que:

- el `map` aplicado a una lista devuelve una lista.
- una lista puede contener repetidos.
- al agregar un elemento a una lista, se agrega al final.
- si un objeto no entiende el mensaje se produce error. Así que sean cuidadosos porque puede pasar de manera intencional.